
3.3.5 Listás vezérlők

1. Egyoszlopos rendezetlen és rendezett lista: lista01
2. Egy- és kétoszlopos lista használata: lista02
3. Egyszeres kiválasztású lista működtetése: lista1
4. Egyszeres és többszörös kiválasztású lista kezelése: lista2
5. Különféle keresési és kiválasztási módok alkalmazása: lista3
6. Egyszeres kiválasztású ablakok szinkronban működtetése: lista4
7. A kijelölt listaelem változtatása jelölőnégyzettel: lista5
8. Egyszerű kombinált lista működése combo1
9. Legördülő kombinált lista működése combo2
10. Lenyíló kombinált lista működése combo3

 Készítsünk programot, amelyben a nevek egyoszlopos listában és rendezetlenül jelennek meg! Legyen
lehetőség a lista elemeinek ábécé sorrendben való megjelenítésére! A kiválasztott listaelemet jelenítsük meg
a szövegmezőben! (lista01)

A LISTA01 alkalmazás űrlapja és a TForm1 osztály:

type
  TForm1 = class(TForm)
    ListBox1: TListBox;
    Edit1: TEdit;
    Allapot: TCheckBox;
    Kilepes: TButton;
    Label1: TLabel;
    procedure FormCreate(Sender: TObject);
    procedure AllapotClick(Sender: TObject);
    procedure KilepesClick(Sender: TObject);
    procedure ListBox1Click(Sender: TObject);
  end;

Deklaráció:

var
  Form1: TForm1;
  nevek : array[1..7] of string;

Eseménykezelő eljárások:

A FormCreate eseménykezelő eljárásban kezdőértéket adunk a nevek tömbnek, mellyel feltöltjük a ListBox1
listát.
procedure TForm1.FormCreate(Sender: TObject);
var
   i : integer;
begin
   nevek[1] := 'László';
   nevek[2] := 'Szilvia';
   nevek[3] := 'Zoltán';
   nevek[4] := 'Zsuzsanna';
   nevek[5] := 'Mónika';
   nevek[6] := 'Tamás';
   nevek[7] := 'Anna';
   // a ListBox1 feltöltése
   for i:=1 to 7 do
         ListBox1.Items.Add(nevek[i]);
end;

A lista elemének kijelöléskor meghívódó ListBox1Click eseménykezelő eljárásban a kijelölt tételt a
szövegmezőbe írjuk.
procedure TForm1.ListBox1Click(Sender: TObject);
begin
  if ListBox1.ItemIndex > -1 then
    // a kijelölt tétel az Edit1 szövegmezőbe írása
    Edit1.Text := ListBox1.Items[ListBox1.ItemIndex];
end;

Az AllapotClick eseménykezelő eljárásban, amely a "Rendezett/Rendezetlen" választógomb kijelölésekor
hívódik meg, töröljük a szövegmező és a listaablak tartalmát. Ha a választógomb ki van jelölve, vagyis
"Rendezett", akkor a ListBox1 Sorted tulajdonságát true-ra állítjuk, és a nevek tömbbel a listát újra
feltöltjük, amely már rendezett lesz. Ha a választógomb nincs kijelölve ("Rendezetlen"), akkor a ListBox1
Sorted tulajdonságát false-ra állítjuk, így az adatok újratöltésekor ismét az eredeti sorrendben jelennek meg.

procedure TForm1.AllapotClick(Sender: TObject);
var
  i : integer;
begin
   // a lista törlése
   ListBox1.Clear;
   // szövegmező törlése
   Edit1.Clear;
   if Allapot.State = cbChecked then
   begin
       // a rendezett állapot beállítása
       ListBox1.Sorted := true;
       // a ListBox1 lista feltöltése
       for i:=1 to 7 do
         ListBox1.Items.Add(nevek[i]);
   end
   else
   begin
      // rendezetlen, eredeti állapot beállítása
      ListBox1.Sorted := false;
      // a ListBox1 lista feltöltése
      for i:=1 to 7 do
         ListBox1.Items.Add(nevek[i]);
   end;
end;

 Módosítsuk a LISTA01 alkalmazást, hogy a lista elemei ábécé sorrendben és egy oszlopban, illetve
rendezetlenül és két oszlopban jelenjenek meg! (lista02)

Változtassunk a deklaráción, hogy több elemet tölthessünk a listára!
var
  Form1: TForm1;
  nevek : array[1..10] of string;
  db : integer;

Eseménykezelő eljárások:

A form betöltésekor hívódó eseménykezelő eljárásban a listaablak feltöltéséhez – a lista01 programhoz
hasonlóan - az Add metódust használjuk, azonban több elemet töltünk be:

procedure TForm1.FormCreate(Sender: TObject);
var
   i : integer;
begin
   // kezdeti értékek beállítása
   db := 10;
   nevek[1] := 'László';
   nevek[2] := 'Szilvia';
   nevek[3] := 'Zoltán';
   nevek[4] := 'Zsuzsanna';
   nevek[5] := 'Mónika';
   nevek[6] := 'Tamás';
   nevek[7] := 'Anna';
   nevek[8] := 'Tibor';
   nevek[9] := 'Margit';
   nevek[10] := 'Szilárd';
   // a ListBox1 lista feltöltése
   for i:=1 to db do
      ListBox1.Items.Add(nevek[i]);
end;

A "Rendezett - 1 oszlopos" választógomb kijelölésekor a Sorted tulajdonság true értékre állításával a nevek
ábécé sorrendben jelennek meg, és egy oszlopban (Columns értéke 1). Ha a választógomb nincs kijelölve
"Rendezetlen - 2 oszlopos", akkor a lista Column tulajdonságát 2-re, a Sorted tulajdonságot pedig false-ra
állítjuk és így töltjük fel a listát.
procedure TForm1.AllapotClick(Sender: TObject);
var
    i : integer;
begin

   ListBox1.Clear;
   Edit1.Clear;
   if Allapot.State = cbChecked then
   begin
       // a lista egy oszlopos lesz
       ListBox1.Columns := 1;
       ListBox1.Sorted := true;
       for i:=1 to db do
         ListBox1.Items.Add(nevek[i]);
   end
   else
   begin
      // lista kétoszlopos
      ListBox1.Columns := 2;
      ListBox1.Sorted := false;
      for i:=1 to db do
        ListBox1.Items.Add(nevek[i]);
   end;
end;

A listaelemen való kattintás megfelel a lista01-nek.

A LISTA02 alkalmazás futási képei:

 Írjunk olyan programot, amely tartalmaz egy egyszeres kiválasztású listaablakot, amelyik alkalmas
elemek hozzáadására a listához, -törlésére, -beszúrására, -helyettesítésére és a teljes lista törlésére! (lista1)

A példánkban nyomógombok vezérlik a listaablak működését. A "Tétel" szövegmezőbe írt szöveg a
"Hozzáad" nyomógomb hatására a lista végéhez kapcsolódik. A "Beszúr" nyomógomb a beírt szöveget a
kijelölt tétel elé szúrja be. A "Tételtörlés" gomb a kiválasztott tételt, míg a "Törlés" nyomógomb a lista
teljes tartalmát törli. Az űrlap:

A TForm1 osztálya:

type
  TForm1 = class(TForm)
    Label1: TLabel;
    Tetel: TEdit;
    Label2: TLabel;
    ListBox1: TListBox;
    Beszur: TButton;
    Teteltorles: TButton;
    Torles: TButton;
    Kilepes: TButton;
    HozzaAd: TButton;
    KivalasztasTorles: TButton;
    Feltolt: TButton;
    Helyettesit: TButton;
    procedure KilepesClick(Sender: TObject);
    procedure TorlesClick(Sender: TObject);
    procedure TeteltorlesClick(Sender: TObject);
    procedure HozzaAdClick(Sender: TObject);
    procedure BeszurClick(Sender: TObject);
    procedure ListBox1Click(Sender: TObject);
    procedure KivalasztasTorlesClick(Sender: TObject);
    procedure FeltoltClick(Sender: TObject);
    procedure HelyettesitClick(Sender: TObject);
  end;

Eseménykezelő eljárások:

A "Hozzáad" nyomógomb megnyomásakor a HozzaAdClick eseménykezelő eljárás hívódik meg, amely a
szövegmezőben megadott szöveget a listaablakba helyezi. Azonos tétel többszöri betételét nem engedjük
meg. Ha a keresett tétel nincs a listában, akkor a tétel mindig a lista végére kerül.
procedure TForm1.HozzaAdClick(Sender: TObject);
begin
   if Tetel.GetTextlen > 0 then
   begin
     if ListBox1.Items.IndexOf(Tetel.text) = -1 then
       ListBox1.Items.Add(Tetel.Text);
   end;
   Tetel.Clear;
   Tetel.SetFocus;
end;

A "Beszúr" nyomógomb megnyomásakor végzett tevékenység eltér a "Hozzáad" nyomógombétól. A
behelyezendő tételt nem a lista végére, hanem a kiválasztott tétel elé szúrja. Ha nincs kiválasztott tétel,
akkor az új tétel a lista elejére kerül.
procedure TForm1.BeszurClick(Sender: TObject);
begin
  if Tetel.GetTextLen > 0 then
  begin
    if ListBox1.Items.IndexOf(Tetel.Text) = -1 then
    begin
      if ListBox1.ItemIndex > -1 then
        // ha van kijelölt tétel, akkor a tétel elé
        ListBox1.Items.Insert(ListBox1.ItemIndex,Tetel.Text)
      else
        // ha nincs kijelölt tétel, akkor az első tétel elé
        ListBox1.Items.Insert(0,Tetel.Text);
     end;
     Tetel.Clear;
     Tetel.SetFocus;
   end;
end;

A "Tételtörlés" nyomógomb megnyomásakor meghívódó TeteltorlesClick eseménykezelő eljárásban, ha van
kiválasztott tétel (ItemIndex > -1 ), azt a Delete metódus hívásával töröljük.
procedure TForm1.TeteltorlesClick(Sender: TObject);
begin
   if ListBox1.ItemIndex > -1 then
     ListBox1.Items.Delete(ListBox1.ItemIndex);
    Tetel.Clear;
    Tetel.SetFocus;
end;

A "Törlés" nyomógomb megnyomásakor a listaablak teljes tartalmát a Clear metódussal töröljük, a fókuszt a
"Tétel" szövegmezőre állítjuk, és engedélyezzük a "Feltölt" nyomógomb működését.
procedure TForm1.TorlesClick(Sender: TObject);
begin
   ListBox1.Clear;
   Tetel.SetFocus;
   Feltolt.Enabled := true;
end;

A listaelemen való kattintás eseménykezelő eljárásában a kijelölt tételt a "Tétel" szövegmezőbe írjuk.
procedure TForm1.ListBox1Click(Sender: TObject);
begin
   if ListBox1.ItemIndex > -1 then
      Tetel.Text := ListBox1.Items[ListBox1.ItemIndex];
end;

A "Kiválasztás megszüntetése" nyomógomb eseménykezelő eljárásában megszüntetjük a kijelölést.
procedure TForm1.KivalasztasTorlesClick(Sender: TObject);
begin
  if ListBox1.ItemIndex > -1 then
     ListBox1.ItemIndex := -1;
end;

A "Feltölt" nyomógomb megnyomásakor meghívódó FeltoltClick eseménykezelő eljárásban feltöltjük a
listát, a Feltolt nyomógombot pedig érvénytelenítjük az Enabled tulajdonság false értékre állításával.
procedure TForm1.FeltoltClick(Sender: TObject);
var
  i : integer;
begin
   for i:=1 to 6 do
   begin
     // lista feltöltése
     ListBox1.Items.Add(IntToStr(i)+'.tétel');
   end;
   Feltolt.Enabled := false;
end;

A "Helyettesít" nyomógomb eseménykezelő eljárásában a kiválasztott tételt a módosított tétellel felülírjuk
oly módon, hogy a kiválasztott tételt a Delete metódussal töröljük, majd a módosított tételt az Insert
metódussal ugyanarra a helyre beszúrjuk.
procedure TForm1.HelyettesitClick(Sender: TObject);
var
    j : integer;
begin
   j := ListBox1.ItemIndex;
   if (j > -1) and (Tetel.GetTextlen > 0) then
   begin
       ListBox1.Items.Delete(j);
       // a módosított tétel a kijelölt helyére kerül
       ListBox1.Items.Insert(j,Tetel.Text)
    end;
end;

A LISTA1 alkalmazás egy futási képe:

 Tervezzünk egy többszörös kiválasztású- és egy egyszeres kiválasztású listaablak kezelésére alkalmas
programot, ahol a két listaablak egymással kommunikál! (lista2)

A "Tételek 1" listaablakban több elemet is kiválaszthatunk, míg a "Tételek 2" listaablakban egyszerre csak
egy tétel kiválasztására van lehetőség. Ezen felül a "Tételek 1" lista elemei nem rendezettek, míg a "Tételek
2" lista elemei névsor szerint rendezettek.

A TForm1 osztálya és az űrlap:

type
  TForm1 = class(TForm)
    Label1: TLabel;
    Label2: TLabel;
    Label3: TLabel;
    Label4: TLabel;
    Label5: TLabel;
    Label6: TLabel;
    Label7: TLabel;
    Tetel: TEdit;
    Tetelek1: TListBox;
    HozzaAd: TButton;
    Osszes: TButton;
    Valaszt: TButton;
    Torles2: TButton;
    Kilepes: TButton;
    Tetelek2: TListBox;
    Torles1: TButton;
    procedure HozzaAdClick(Sender: TObject);
    procedure Torles1Click(Sender: TObject);
    procedure Torles2Click(Sender: TObject);
    procedure OsszesClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
    procedure ValasztClick(Sender: TObject);
    procedure KilepesClick(Sender: TObject);
    procedure Tetelek1Click(Sender: TObject);
  end;

Eseménykezelő eljárások:

A FormCreate eseménykezelő eljárásban "Tételek 1" listára vonatkozó többszörös kiválasztást jelentő
MultiSelect tulajdonságot true-ra, a "Tételek 2" lista esetén pedig az egyszeres kiválasztást jelentő false
értékre állítjuk. A "Tételek 1" lista tételei nem rendezettek. (a Sorted tulajdonsága false, – ami az
alapértelmezés). A "Tételek 1" listának kezdőértéket adunk. A "Tételek 2" lista elemei rendezettek lesznek,
mivel a Sorted tulajdonságot true-ra állítottuk.
procedure TForm1.FormCreate(Sender: TObject);
begin
   // a "Tételek 1" lista esetén a többszörös elemkiválasztás engedélyezése
   Tetelek1.MultiSelect := true;
   Tetelek1.ExtendedSelect := true;
   // a "Tételek 1" lista rendezetlen
   Tetelek1.Sorted := false;
   // a "Tételek 1" lista kezdeti feltöltése
   Tetelek1.Items.Add('D. tétel');
   Tetelek1.Items.Add('X. tétel');
   Tetelek1.Items.Add('C. tétel');
   Tetelek1.Items.Add('A. tétel');
   Tetelek1.Items.Add('K. tétel');
   // a "Tételek 2" lista egyszeres kiválasztású
   Tetelek2.MultiSelect := false;
   // a "Tételek 2" lista rendezett
   Tetelek2.Sorted := true;
end;

A "Hozzáad" nyomógomb megnyomásakor a HozzaAdClick eseménykezelő eljárás hívódik meg, amely a
szövegmezőben megadott szöveget a "Tételek 1" listaablakba helyezi. Az eljárás csak olyan tételt helyez a
listába, amely még nem szerepelt benne. Az eseménykezelő törli a "Tétel" szövegmező tartalmát és ráhelyezi
a fókuszt.
procedure TForm1.HozzaAdClick(Sender: TObject);
begin
   if Tetel.GetTextLen > 0 then
     if Tetelek1.Items.Indexof(Tetel.Text) = -1 then
        Tetelek1.Items.Add(Tetel.Text);
   // a szövegmező törlése
   Tetel.Clear;
   Tetel.SetFocus;
end;

Az "Összes>>" nyomógomb megnyomásakor az OsszesClick eseménykezelő eljárás hívódik meg, amely a
"Tételek 1" lista teljes tartalmát átmásolja a "Tételek 2" listába, majd a fókuszt a "Tétel" szövegmezőre
helyezi.
procedure TForm1.OsszesClick(Sender: TObject);
var
    i : integer;
begin
   Tetelek2.Clear;
   for i:=0 to Tetelek1.Items.Count-1 do
     // az összes tétel átmásolásra kerül
     Tetelek2.Items.Add(Tetelek1.Items[i]);
   // a szövegmező törlése
   Tetel.Clear;
  // a fókusz átállítása a szövegmezőre
   Tetel.SetFocus;
end;

Az "Választ>>" nyomógomb megnyomásakor a ValasztClick eseménykezelő eljárás hívódik meg, amely a
"Tételek 1" lista kiválasztott elemeit átmásolja a "Tételek 2" listába, majd a fókuszt a "Tétel" szövegmezőre
helyezi.
procedure TForm1.ValasztClick(Sender: TObject);
var
   i : integer;
begin
   // a "Tételek 2" lista törlése
   Tetelek2.Clear;
   for i:=0 to Tetelek1.Items.Count-1 do
     if Tetelek1.Selected[i] then
     // a "Tételek 1" lista kiválasztott tételei
     // a "Tételek 2" listában is megjelennek
       Tetelek2.Items.Add(Tetelek1.Items[i]);
   Tetel.Clear;
   // a fókusz átállítása a szövegmezőre
   Tetel.SetFocus;
end;

A "<<Törlés" nyomógomb megnyomásakor a "Tételek 1" lista teljes tartalmát töröljük, és a fókuszt a "Tétel"
szövegszerkesztőre állítjuk.
procedure TForm1.Torles1Click(Sender: TObject);
begin
    Tetelek1.Clear;
    // a szövegmező törlése
    Tetel.Clear;
    // a fókusz átállítása a szövegmezőre
    Tetel.SetFocus;
end;

A "Törlés >>" nyomógomb megnyomásakor a "Tételek 2" lista teljes tartalmát töröljük a Clear metódussal
és ráállítjuk a fókuszt a "Tétel" szövegmezőre.
procedure TForm1.Torles2Click(Sender: TObject);
begin
   Tetelek2.Clear;
   // a szövegmező törlése
   Tetel.Clear;
   // a fókusz átállítása a szövegmezőre
   Tetel.SetFocus;
end;

A futási kép:

Használjunk különféle keresési és kiválasztási módokat a többszörös kiválasztású listaablakban!
(lista3)

A "Keres" és a "Keres pontosan" nyomógombokkal a "Tétel" adatbeviteli mezőbe írt szöveget keressük a
"Tételek 2" listában. A "=Kiválaszt=>", illetve a "Megszüntet=>" nyomógombokkal a "Tételek 1" listában
kiválasztott elemeket kiválasztjuk, illetve kiválasztásukat megszüntetjük a "Tételek 2" listában. Az
"Összes=>", illetve "Egyik sem=>" nyomógombokkal kiválasztjuk a "Tételek 2" lista összes elemét, illetve
megszüntetjük a kiválasztásukat. A "Hozzáad" nyomógombbal a "Tétel" szövegmezőbe írt szöveget mindkét
listához hozzáadjuk.

A TForm1 osztálya és az űrlap:
type
  TForm1 = class(TForm)
    Label1: TLabel;
    Label2: TLabel;
    Label8: TLabel;
    Tetelek1: TListBox;
    Tetel: TEdit;
    HozzaAd: TButton;
    Kivalaszt: TButton;
    Megszuntet: TButton;
    Osszes: TButton;
    EgyikSem: TButton;
    Torles1: TButton;
    Kilepes: TButton;
    Tetelek2: TListBox;
    KeresPontosan: TButton;
    Keres: TButton;
    Torles2: TButton;
    Feltoltes: TButton;
    procedure KilepesClick(Sender: TObject);
    procedure HozzaAdClick(Sender: TObject);
    procedure KivalasztClick(Sender: TObject);
    procedure MegszuntetClick(Sender: TObject);
    procedure OsszesClick(Sender: TObject);
    procedure EgyikSemClick(Sender: TObject);
    procedure KeresClick(Sender: TObject);
    procedure KeresPontosanClick(Sender: TObject);
    procedure Torles1Click(Sender: TObject);
    procedure FormCreate(Sender: TObject);
    procedure Tetelek1Click(Sender: TObject);
    procedure Torles2Click(Sender: TObject);
    procedure FeltoltesClick(Sender: TObject);
  private
  procedure Feltolt(lst: TListBox);
end;

Eseménykezelő eljárások:

A FormCreate eseménykezelő eljárásban beállítjuk az ablak bal felső sarokpontjának koordinátáit, valamint
alapállapotként néhány elemet teszünk a "Tételek 1" és a "Tételek 2" listákba. A "Tételek 1" és a "Tételek 2"
listák többszörös elemkiválasztást engedélyező MultiSelect és az ExtededSelect tulajdonságokat mindkét
esetben true értékre állítjuk.

procedure TForm1.FormCreate(Sender: TObject);
begin
  // többszörös tételkiválasztás engedélyezése
  Tetelek1.MultiSelect    := true;
  Tetelek1.ExtendedSelect := true;
  Tetelek2.MultiSelect    := true;
  Tetelek2.ExtendedSelect := true;
  // a "Tételek 1" lista kezdeti feltöltése
  Feltolt(Tetelek1);
  // a "Tételek 2" lista kezdeti feltöltése
  Feltolt(Tetelek2);
end;

A Feltolt eljárás a TForm1 osztály private eseménykezelő eljárása, amely paraméterként átveszi a
feltöltendő lista objektumot.
procedure TForm1.Feltolt(lst: TListBox);
var
    i : integer;
begin
   with lst do
   begin
     for i:=1 to 4 do
       Items.Add(chr(48+i)+'. tétel');
     Items.Add('alma');
     Items.Add('ananász');
     Items.Add('stb.');
   end;
end;

A "Hozzáad" nyomógomb megnyomásakor a HozzaAdClick eseménykezelő eljárás hívódik meg, amely a
"Tétel" szövegmezőben megadott szöveget a "Tételek 1" és a "Tételek 2" listákba helyezi, ha az elem még
nem létezik, továbbá törli a "Tétel" szövegmező tartalmát, és a fókuszt ráhelyezi.
procedure TForm1.HozzaAdClick(Sender: TObject);
begin
  if Tetel.GetTextLen > 0 then   // a szövegmező nem üres
  begin
    if Tetelek2.Items.IndexOf(Tetel.Text) = -1 then
    // az új tétel hozzáadása a "Tételek 2" listához
       Tetelek2.Items.Add(Tetel.Text);
    if Tetelek1.Items.IndexOf(Tetel.Text) = -1 then
      // az új tétel hozzáadása a "Tételek 1" listához
       Tetelek1.Items.Add(Tetel.Text);
  end;
  // a "Tétel" szövegmező törlése
  Tetel.Clear;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A "Kiválaszt" nyomógomb megnyomásakor meghívódó KivalasztClick eseménykezelő eljárásban először
megszüntetjük a "Tételek 2" listában a kiválasztásokat. Ezt követően a "Tételek 1" lista kiválasztott tételeit
egyenként összehasonlítjuk a "Tételek 2" lista összes elemével, egyezés esetén a tételt kiválasztottá tesszük a
"Tételek 2" listaablakban.
procedure TForm1.KivalasztClick(Sender: TObject);
var
     i, j : integer;
begin
   // a "Tételek 2" listában a kiválasztások megszüntetése
   for j:=0 to Tetelek2.Items.Count-1 do
     Tetelek2.Selected[j] := False;
   for i:=0 to Tetelek1.Items.Count-1 do
   begin
     if Tetelek1.Selected[i] then
     begin
       // a "Tételek 1" lista i-edik tétele van kiválasztva
       for j:=0 to Tetelek2.Items.Count-1 do
       begin
         if Tetelek1.Items[i] = Tetelek2.Items[j] then
           // a "Tételek 2" lista j-edik tételét választja ki
           Tetelek2.Selected[j] := True;
       end;
     end;
   end;
   // a szövegmező törlése
   Tetel.Clear;
   // a fókusz átállítása a szövegmezőre
   Tetel.SetFocus;
end;

A "Tételek 1" listaablak valamelyik elemén kattintva meghívódik a Tetelek1Click eseménykezelő eljárás,
amelyben a megjelölt listaelem a "Tétel" szövegmezőbe kerül:
procedure TForm1.Tetelek1Click(Sender: TObject);
var
   i : integer;
begin
   for i:=0 to Tetelek1.Items.Count-1 do
   begin
     if Tetelek1.Selected[i] then
     begin
        // a kiválasztott tétel a "Tétel" szövegmezőbe
        // kerül
        Tetel.Text := Tetelek1.Items[i];
        break;
     end;
   end;
end;

A "Megszüntet" nyomógomb megnyomásakor meghívódó MegszuntetClick eseménykezelő eljárásban
megszüntetjük az elemek kiválasztását a "Tételek 2" listában.
procedure TForm1.MegszuntetClick(Sender: TObject);
var
    i : integer;
begin
  for i:=0 to Tetelek2.Items.Count-1 do
    Tetelek2.Selected[i] := false;
   // a fókusz átállítása a szövegmezőre
   Tetel.SetFocus;
end;

A "Összes" nyomógomb megnyomásakor meghívódó OsszesClick eseménykezelő eljárásban a "Tételek 2"
lista összes elemét kiválasztjuk:
procedure TForm1.OsszesClick(Sender: TObject);
var
   i : integer;
begin
  // a "Tételek 2" lista összes elemét kiválasztja
  for i:=0 to Tetelek2.Items.Count-1 do
     Tetelek2.Selected[i] := True;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A "Egyik sem" nyomógomb megnyomásakor meghívódó EgyikSemClick eseménykezelő eljárásban a
"Tételek 2" listában a tételek kiválasztását megszüntetjük.
procedure TForm1.EgyikSemClick(Sender: TObject);
var
   i : integer;
begin
  // a "Tételek 2" lista összes elemének
  // kiválasztását megszünteti
  for i:=0 to Tetelek2.Items.Count-1 do
     Tetelek2.Selected[i] := False;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A "Törlés1" nyomógomb megnyomásakor a "Tételek 2" lista teljes tartalmát a Clear metódussal töröljük, és
a fókuszt a "Tétel" szövegmezőre állítjuk.
procedure TForm1.Torles1Click(Sender: TObject);
begin
  // a "Tételek 1" lista törlése
  Tetelek1.Clear;
  // a "Tétel" szövegmező törlése
  Tetel.Clear;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A "Törlés 2" nyomógomb megnyomásakor a "Tételek 2" lista teljes tartalmát a Clear metódussal töröljük, és
a fókuszt a "Tétel" szövegmezőre állítjuk.
procedure TForm1.Torles2Click(Sender: TObject);
begin
  // a "Tételek 2" lista törlése
  Tetelek2.Clear;
  // a "Tétel" szövegmező törlése
  Tetel.Clear;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A "Keres" nyomógomb megnyomásakor a "Tétel" adatbeviteli mezőbe beírt szöveggel kezdődő tételt a
"Tételek 2" listában keressük. Sikeres keresés után a megfelelő tételt kiválasztjuk.

procedure TForm1.KeresClick(Sender: TObject);
var
   i   : integer;
begin
  for i:=0 to Tetelek2.Items.Count-1 do
  begin
     // összes kiválasztás megszüntetése
     // a "Tételek 2" listában
     Tetelek2.Selected[i] := False;
  end;
  // tétel keresése
  for  i:=0 to Tetelek2.Items.Count-1 do
  begin
     if pos(Tetel.Text,Tetelek2.Items[i]) = 1 then
     begin
       // a szövegmező tartalmával
       // kezdődő tétel kijelölése
       Tetelek2.Selected[i] := True;
     end;
  end;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A "Keres pontosan" nyomógomb megnyomásakor a "Tétel" adatbeviteli mezőbe írt tételt a "Tételek 2"
listában keressük. Először megszüntetjük a kiválasztásokat a "Tételek 2" listában, majd a keresett tételt
összehasonlítjuk a lista összes elemével. Sikeres keresés esetén a listaelemet kiválasztottá tesszük.
procedure TForm1.KeresPontosanClick(Sender: TObject);
var
   j : integer;
begin
  // a szövegmezőben levő tétel keresése
  for j:=0 to  Tetelek2.Items.Count-1 do
    if Tetel.Text = Tetelek2.Items[j] then
       // a megtalált tétel kijelölése
       Tetelek2.Selected[j] := True
       else
       // kiválasztás megszüntetése
        Tetelek2.Selected[j] := False;
  // a fókusz átállítása a szövegmezőre
  Tetel.SetFocus;
end;

A LISTA3 alkalmazás futási ablaka:

 Működtessünk szinkronban egyszeres kiválasztású listaablakokat! (lista4)

A "Tételek hossza" lista a "Tételek" lista elemeinek karakterszámát tartalmazza. Bármelyik listában történő
választás hatására a megfelelő elem a másikban is kiválasztásra kerül. A "Kiválaszt" nyomógomb meg-
nyomásakor a "Tétel" szövegmezőben található szöveggel kezdődő első listaelemet kiválasztottá tesszük
mindkét listában.

A TForm1 osztálya és az űrlap:
type
    TForm1 = class(TForm)
    Label1: TLabel;
    Label2: TLabel;
    Label3: TLabel;
    Tetel: TEdit;
    Tetelek: TListBox;
    HozzaAd: TButton;
    Kivalaszt: TButton;
    TetelekHossza: TListBox;
    Torles: TButton;
    Kilepes: TButton;
    procedure TorlesClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
    procedure KilepesClick(Sender: TObject);
    procedure HozzaAdClick(Sender: TObject);
    procedure KivalasztClick(Sender: TObject);
    procedure TetelekClick(Sender: TObject);
    procedure TetelekHosszaClick(Sender: TObject);
  private
    procedure TetelekHosszaBeiras;
  end;

A "Tételek hossza" lista feltöltését a TetelekHosszaBeiras metódussal végezzük. Az eljárás minden egyes
hívásakor először törli a "Tételek hossza" lista tartalmát, majd kiszámítja a "Tételek" lista elemeinek hosszát
(length), amelyet sztringgé alakítva (IntToStr) hozzáad a "Tételek hossza" listához.
procedure TForm2.TetelekHosszaBeiras;
var
   i, h : integer;
   buff : string[30];
begin
 // a "Tételek hossza" listaablak tartalmának törlése
  TetelekHossza.Clear;
  for i:=0 to Tetelek.Items.Count-1 do
  begin
    // a "Tételek" listaablak tételhosszának átalakítása sztringgé
    buff := IntToStr(length(Tetelek.Items[i]));
    // a "Tételek hossza" listaablakba íródik a buff tartalma
    TetelekHossza.Items.Add(buff);
  end;
end;

Eseményekezelő eljárások:

A form betöltésekor meghívódó FormCreate eseménykezelő eljárásban kezdőértéket adunk a "Tételek"
listának, majd meghívjuk a TetelekHosszaBeiras eljárást a "Tételek hossza" lista feltöltésére.
procedure TForm1.FormCreate(Sender: TObject);
begin
   // a "Tételek" lista feltöltése
   Tetelek.Items.Add('alma');
   Tetelek.Items.Add('dió');
   Tetelek.Items.Add('banán');
   Tetelek.Items.Add('eper');
   Tetelek.Items.Add('meggy');
   Tetelek.Items.Add('cseresznye');
   TetelekHosszaBeiras;
end;



18

A "Törlés" nyomógomb megnyomásakor mindkét lista tartalmát a töröljük.
procedure TForm1.TorlesClick(Sender: TObject);
begin
   // a "Tételek" lista törlése
   Tetelek.Clear;
   // a "Tételek hossza" lista törlése
   TetelekHossza.Clear;
end;

A "Hozzáad" nyomógomb megnyomásakor a HozzaAdClick eseménykezelő eljárás hívódik meg, először
törli mindkét lista kiválasztását, majd a "Tétel" szövegmezőben megadott szöveget a "Tételek" listába
helyezi, ha a tétel még nincs benne. Ezt követően meghívja a TetelekHosszaBeiras eljárást, hogy frissítse az
új tétellel a "Tételek hossza" listaablakot. Végezetül törli "Tétel" adatbeviteli mezőt és ráhelyezi a fókuszt.
procedure TForm1.HozzaAdClick(Sender: TObject);
var
   j: integer;
begin
   // mindkét listába a kiválasztások törlése
   for j:=1 to TetelekHossza.Items.Count-1 do
   begin
     TetelekHossza.Selected[j]:=false;
     Tetelek.Selected[j]:= false;
   end;
  // a szövegmező tartalmának vizsgálata
  if Tetel.GetTextLen > 0 then
  begin
     if Tetelek.Items.IndexOf(Tetel.Text) = -1 then
     begin
       Tetelek.Items.Add(Tetel.Text);
       // a "Tételek hossza" lista törlése
       TetelekHosszaBeiras;
     end;
  end;
  // a szövegmező tartalmának törlése
  Tetel.Clear;
  Tetel.SetFocus;
end;

A "Kiválaszt" nyomógomb megnyomásakor meghívódó KivalasztClick eljárásban először mindkét listában
megszüntetjük a kiválasztást. Ezt követően megnézzük, hogy a "Tétel" szövegmezőben megadott szöveggel
kezdődő elem szerepel-e a "Tételek" listában. Ha megtaláltunk a tételt, akkor mindkét listában kiválasztjuk a
megfelelő elemet:
procedure TForm1.KivalasztClick(Sender: TObject);
var
   i, j : integer;
begin
   // mindkét listában töröljük a kiválasztást
   for j:=1 to TetelekHossza.Items.Count-1 do
   begin
     TetelekHossza.Selected[j]:=false;
     Tetelek.Selected[j]:= false;
   end;
   if Tetel.GetTextLen > 0 then
   begin
     i:=Tetelek.Items.IndexOf(Tetel.Text);
     if i <> -1 then
     begin
       // a keresett tétel benne van a listában
       Tetelek.Selected[i]:= true;
       TetelekHossza.Selected[i]:= true;
     end;
     Tetel.SetFocus;
   end
end;

A "Tételek" listaablak valamelyik elemén kattintva meghívódik a TetelekClick eseménykezelő eljárás, ahol a
"Tételek" listában kiválasztott tételnek megfelelő sorszámú elem a "Tételek hossza" listában is kiválasztásra
kerül.
procedure TForm1.TetelekClick(Sender: TObject);
begin
   if TetelekHossza.ItemIndex >= 0 then
     TetelekHossza.Selected[TetelekHossza.ItemIndex] := false;
   if Tetelek.ItemIndex >= 0 then
     TetelekHossza.Selected[Tetelek.ItemIndex] := true;
end;

A "TételekHossza" lista valamelyik tételén történő kattintás hatására meghívódó TetelekHosszaClick
eseménykezelő eljárásban a "TételekHossza" listában kiválasztott elemnek megfelelő sorszámú elem a
"Tételek" listában is kiválasztottá válik.

procedure TForm1.TetelekHosszaClick(Sender: TObject);
begin
 if Tetelek.ItemIndex >= 0 then
     Tetelek.Selected[Tetelek.ItemIndex] := false;
 if TetelekHossza.ItemIndex >= 0 then
     Tetelek.Selected[TetelekHossza.ItemIndex] := true;
end;

Az 'eper' szöveg keresése a "Kiválaszt" nyomógomb hatására:

 Változtassuk meg a lista kijelölt elemét jelölőnégyzet segítségével! (lista5)

Ha a "Tételek" lista "be" elemén kattintunk, a jelölőnégyzet a tételnek megfelelő bekapcsolt állapotot
mutatja. Ha azonban a jelölőnégyzeten kattintunk az egérrel, a jelölőnégyzet kikapcsolt állapotba kerül, és a
lista eleme "ki"-re változik.

A TForm1 osztálya és az űrlap:
type
  TForm1 = class(TForm)
    Label1: TLabel;
    Label2: TLabel;
    Label3: TLabel;
    Label4: TLabel;
    Label5: TLabel;
    Label6: TLabel;
    ListBox1: TListBox;
    Valaszt: TCheckBox;
    Kilepes: TButton;
    procedure ValasztClick(Sender: TObject);
    procedure KilepesClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
    procedure ListBox1Click(Sender: TObject);
end;

Eseménykezelő eljárások:

A form betöltésekor meghívódó a FormCreate eseménykezelő eljárásban a "Tételek" listának kezdőértéket
adunk, és beállítjuk az ablak bal felső sarkának koordinátáit.
procedure TForm1.FormCreate(Sender: TObject);
var
   i : integer;
begin
   randomize;
   // a "Tételek" listaablak feltöltése
   for i:=1 to 10 do
   if random(1998) mod 2 = 1 then
      ListBox1.Items.Add('be')
   else
     ListBox1.Items.Add('ki');
end;

A "Tételek" lista elemein való kattintáskor meghívódó ListBox1Click eseménykezelő eljárásban a
kiválasztott tételnek megfelelő állapotra állítjuk a jelölőnégyzetet. A "be" tétel esetén a jelölőnégyzet State
tulajdonságát cbChecked, "ki" tétel esetén pedig cbUnchecked értékre állítjuk.
procedure TForm1.ListBox1Click(Sender: TObject);
var
   i : integer;
begin
  for i:=0 to ListBox1.Items.Count-1 do
  begin
     if ListBox1.Selected[i] then
     begin
       if ListBox1.Items[i] = 'be' then
         // a jelölőnégyzetet jelölt állapotba állítja
         Valaszt.State := cbChecked
       else
         // a jelölőnégyzetet jelöletlen állapotba állítja
         Valaszt.State := cbUnchecked;
     end;
  end;
end;



21

A "Be/Ki" jelölőnégyzeten való kattintás hatására meghívódó ValasztClick eseménykezelő eljárás a
"Tételek" listában megkeresi a kiválasztott tételt és törli azt. Ezután a jelölőnégyzet állapotának megfelelő
elemet visszaírja a listaablakba a törölt elem helyére, és a kiválasztást is a tételre helyezi.

procedure TForm1.ValasztClick(Sender: TObject);
var
   i : integer;
begin
  for i:=0 to ListBox1.Items.Count-1 do
  begin
     if ListBox1.Selected[i] then
     // az i-edik tétel kiválasztott
     begin
       // a kiválasztott tétel törlése
       ListBox1.Items.Delete(i);
       if Valaszt.State = cbChecked then
         // a jelölőnégyzet bejelölt, ezért
         // a listaablakba a 'be' szöveget írjuk
         ListBox1.Items.Insert(i,'be')
       else
         // a jelölőnégyzet nincs kijelölve
         // a listaablakba a 'ki' szöveg kerül
         ListBox1.Items.Insert(i,'ki');
         // i-edik tétel kijelölése
       ListBox1.Selected[i] := true;
       break;
     end;
   end;
end;

A LISTA5 alkalmazás futási képei:

 Készítsünk programot az egyszerű kombinált lista működésének bemutatására! Legyen lehetőség elem
hozzáadására, -törlésére és a lista tartalmának törlésére! (combo1)

A TForm1 osztálya és az űrlap:

Egyszerű kombinált lista feltölthető listával, illetve egy szövegszer-
kesztővel elemenként.
type
  TForm1 = class(TForm)
    ComboBox1: TComboBox;
    HozzaAd: TButton;
    Teteltorles: TButton;
    Torles: TButton;
    Feltolt: TButton;
    Kilepes: TButton;
    Label1: TLabel;
    Olvas: TButton;
    Tetel: TEdit;
    Label2: TLabel;
    procedure FeltoltClick(Sender: TObject);
    procedure KilepesClick(Sender: TObject);
    procedure TeteltorlesClick(Sender: TObject);
    procedure TorlesClick(Sender: TObject);
    procedure HozzaAdClick(Sender: TObject);
    procedure OlvasClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
  private
    { Private declarations }
  public
    { Public declarations }
  end;

Eseménykezelő eljárások:

Indításkor beállítjuk, hogy egyszerű kombinált listát használunk
procedure TForm1.FormCreate(Sender: TObject);
begin
   // a kombinált lista stílus beállítása
   ComboBox1.Style := csSimple;
end;

A Feltolt nyomógomb eseménykezelő eljárása
procedure TForm1.FeltoltClick(Sender: TObject);
var
   i : integer;
begin
   for i:= 1 to 6 do
     ComboBox1.Items.Add(IntToStr(i)+'. tétel');
   // a "Feltölt" nyomógomb letiltása
   Feltolt.Enabled := False;
end;

A Kilepes nyomógomb eseménykezelő eljárása
procedure TForm1.KilepesClick(Sender: TObject);
begin
   Application.Terminate;
end;

{ a "Tételtörlés" nyomógomb eseménykezelő eljárása }
procedure TForm1.TeteltorlesClick(Sender: TObject);
begin
   if ComboBox1.ItemIndex > -1 then
     ComboBox1.Items.Delete(ComboBox1.ItemIndex);
   Tetel.Clear;
end;

A Torles nyomógomb eseménykezelő eljárása
procedure TForm1.TorlesClick(Sender: TObject);
begin
  // a "Feltölt" nyomógomb engedélyezése
  Feltolt.Enabled := True;
  ComboBox1.Clear;
end;

A Hozzaad nyomógomb eseménykezelő eljárása
procedure TForm1.HozzaAdClick(Sender: TObject);
begin
   if length(ComboBox1.Text) > 0 then
  begin
   if ComboBox1.Items.Indexof(ComboBox1.Text) = -1 then
   begin
    ComboBox1.Items.Add(ComboBox1.Text);
    ComboBox1.Text := '';
   end;
  end;
end;

Az Olvas nyomógomb eseménykezelő eljárása
procedure TForm1.OlvasClick(Sender: TObject);
begin
   Tetel.Text := ComboBox1.Text;
end;

A program futás közben:

Módosítsuk a COMBO1 feladatban a ComboBox1 vezérlő stílusát úgy, hogy az legördülő kombinált
lista legyen! (combo2)

A combo1 feladathoz képest csak annyi a változás, hogy a ComboBox1 lista stílusa csDropDown.

Az űrlap és a futási kép:

Módosítsuk a COMBO1 feladatban a ComboBox1 vezérlő stílusát úgy, hogy az lenyíló kombinált lista
legyen! (combo3)

A combo1 feladathoz képest csak annyi a változás, hogy a ComboBox1 lista stílusa csDropDownList és a
HozzaAd gombot letíltjuk.

Az űrlap és a futási kép:

  


