

### 3.5 Formok és vezérlők programból történő létrehozása

1. Gombok dinamikus használata [Dingomb](#)
2. Digitális lottószelvény [Lotto](#)



Készítsünk egy alkalmazást, amelyben csúszka mozgatással lehet számozott gombokat létrehozni és a gombnyomás feliratának megfelelően változtatja a form címsorát! (*Dingomb*)

Helyezzünk a formra egy függőleges csúszkát (*Scrollbar1*) és legyen a *Scrollbar1.Min=1* és a *Scrollbar1.Max=10*!

Globálisan deklaráljunk a form moduljában egy dinamikus tömböt, amely a létrehozott gombokat tárolja és egy egészet, a gombok számának tárolására.

```
var
  gomb:array of TButton;
  gombszam : integer;
```

Készítsünk egy olyan eljárást, amelyik megfelelő méretűre állítja a dinamikus tömböt és elkészíti, illetve felszabadítja a számozott gomb tömbelemeket.

```
procedure létrehozas(hanyat:integer; var hanybol:integer);
var i:integer;
begin
  for i:=hanyat to hanybol-1 do // a feleslegeseket felszabadítjuk
  begin
    gomb[i].Free;
  end;
  Setlength(gomb,hanyat); // beállítjuk a dinamikus tömb méretét
  for i:=0 to hanyat-1 do
  begin
    if gomb[i]=nil then // ha nem léteznek létrehozzuk a gombokat
    begin
      gomb[i]:=TButton.Create(Form1);
      gomb[i].parent:=Form1;
      gomb[i].visible:=true;
      gomb[i].Caption:=inttostr(i+1);
      gomb[i].width:=Form1.Scrollbar1.Left;
      gomb[i].height:=Form1.ClientHeight div 10;
      gomb[i].top:=i*gomb[i].height;
      gomb[i].OnClick:=Form1.gombnyomas;
    end;
  end;
  hanybol:=hanyat;
end;
```



Ezek után az ablak létrehozásakor egy gombot hozunk létre.

```
procedure TForm1.FormCreate(Sender: TObject);
begin
  létrehozas(1,gombszam);
end;
```

A csúszka mozgatásakor a pozíciónak megfelelően kezeljük a gombokat.


```
procedure TForm1.Scrollbar1Change(Sender: TObject);
begin
  létrehozas(Scrollbar1.Position,gombszam);
end;
```

Futáskor a csúszkának megfelelő számú gomb látszik, és bármelyik megnyomása-kor a gomb felirata felkerül a form címsorába.





A feladat megoldásához 90 nyomógombot helyezünk el az ablakban. Ilyen nagyszámú vezérlő kezelése csak vezérlőtömb alkalmazásával képzelhető el, a nyomógombok futás közbeni létrehozásával. A nyomógombok elhelyezkedését és működését az alábbi ábra szemlélteti:



```
TLottoFrm = class(TForm)
    SzelvenyBtn: TButton;
    KilepBtn: TButton;
    procedure FormCreate(Sender: TObject);
    procedure BLottoClick(Sender: TObject);
    procedure FormClose(Sender: TObject; var Action: TCloseAction);
    procedure SzelvenyBtnClick(Sender: TObject);
    procedure KilepBtnClick(Sender: TObject);
private
    BLotto : array[0..5,0..14] of TButton;
    db : byte;
    procedure Torol;
end;
```

A program indításakor (*FormCreate*) minden nyomógombot külön létrehozunk a **Create** konstruktor hívásával, melynek paramétere a vezérlő tulajdonosa, jelen esetben a *LottoFrm* formobjektum. Az így létrehozott gomb csak akkor jelenik meg az ablakban, ha a **Parent** tulajdonságot is beállítjuk (*LottoFrm*). A gombokhoz közös eseménykezelő eljárást is készítettünk, melyet mindenképpen metódusként kell definiálni. Az eseménykezelő beállítását az **OnClick** tulajdonság értékének megadásával végeztük el:

```
procedure TLottoFrm.FormCreate(Sender: TObject);
var
    i,j : integer;
begin
    for i:=0 to 5 do
        for j:=0 to 14 do
            begin
                BLotto[i,j]:=TButton.Create(LottoFrm);
                with BLotto[i,j] do
                    begin
                        // a szülőablak
                        parent:=LottoFrm;
                        // a gomb számított helye
                        width:=26;
                        height:=24;
                        left:=LottoFrm.clientwidth div 2 -(8-j)*(width+3)+ 16;
                        top:=LottoFrm.clientheight div 2 -(3-i)*(height+2)+12;
                        // a gomb feliratának megadása
                        Caption:=inttostr(i*15+j+1);
                        // a gomb eseménykezelője
                        OnClick:=BLottoClick;
                    end;
                end;
            // a fontméret beállítása 10 pontra,
            // és Tag tulajdonság nullázása
            Torol;
        end;
end;
```

A nyomógombok közös eseménykezelő eljárásában csak öt szám kijelölését engedjük meg, a számot tartalmazó gomb **Font.Size** és a **Tag** tulajdonságának átállításával. A kijelölt számok jelölése újbóli kattintáskor megszűnik.

```

procedure TLottoFrm.BLottoClick(Sender: TObject);
begin
    with (Sender as TButton) do
        if tag=1 then // kiválasztott volt
            begin
                Font.Size:=10;
                tag:=0;
                dec(db);
            end
        else // nem volt kiválasztva
            begin;
                if db>=5 then
                    begin
                        beep; exit;
                    end;
                Font.Size:=16;
                tag:=1;
                LottoFrm.ActiveControl:=nil;
                inc(db);
            end;
end;

```

Felhívjuk a figyelmet az **As** operátor használatára, amellyel a **Sender** (küldő) paraméter típusát **TButton** típusra módosítjuk. A komponensek **Tag** tulajdonsága a programozó által szabadon használható **longint** típusú tároló. A megoldásban a **Tag** tulajdonság 1 értéke a kiválasztott gombokat jelöli.). Az eseménykezelőben elvégzett beállításokat az "Új szelvény" gomb megnyomásakor töröljük a *Torol* metódus hívásával.

```

procedure TLottoFrm.SzelvenyBtnClick(Sender: TObject);
begin
    Torol;
    db:=0;
end;

// A gombokat kiindulási állapotba állítjuk
procedure TLottoFrm.Torol;
var
    i,j : integer;
begin
    for i:=0 to 5 do
        for j:=0 to 14 do
            with BLotto[i,j] do
                begin
                    tag:=0;
                    Font.Size:=10;
                end;
end;

```

A létrehozás és a használat után nem szabad megfelelkezni a kilépéskor kötelező takarításról sem:

```

procedure TLottoFrm.FormClose(Sender: TObject;
                                var Action: TCloseAction);
var
    i,j : integer;
begin
    for i:=0 to 5 do
        for j:=0 to 14 do
            if Assigned(BLotto[i,j]) then BLotto[i,j].Free;
end;

```